# Handout 19: Movement

**1.** Handling movement

```
% start Root

###  Grammar
NP[f=?f] -> Det[f=?f] N[f=?f]
Root -> S[-gap]
Root -> Aux[f=?f, s=?v] NP[f=?f] VP[f=?v, -gap]
Root -> Wh Aux[f=?f, s=?v] NP[f=?f] VP[f=?v, +gap]
S[gap=?g] -> NP[f=?f] VP[f=?f, gap=?g]
VP[f=?f, -gap] -> V[f=?f, -t, s=0]
VP[f=?f, -gap] -> V[f=?f, +t, s=0] NP
VP[f=?f, +gap] -> V[f=?f, +t, s=0]
VP[f=?f, gap=?g] -> Aux[f=?f, s=?v] VP[f=?v, gap=?g]

###  Lexicon
Det[f=sg] -> 'a'
Det -> 'the'
N[f=sg] -> 'cat' | 'dog'
N[f=pl] -> 'cats' | 'dogs'
Aux[f=sg, s=ing] -> 'is' | 'was'
V[f=sg, -t, s=0] -> 'barks'
V[f=pl, -t, s=0] -> 'bark'
V[f=ing, -t, s=0] -> 'barking'
V[f=sg, +t, s=0] -> 'chases'
V[f=ing, +t, s=0] -> 'chasing'
Wh -> 'who'
```

**2.** Examples

**a.** `the dog is barking`

```
(Root[]
  (S[-gap]
    (NP[f='sg'] (Det[] the) (N[f='sg'] dog))
    (VP[f='sg', -gap]
      (Aux[f='sg', s='ing'] is)
      (VP[f='ing', -gap] (V[f='ing', s=0, -t] barking)))))
```

**b.** `is the dog barking`

```
(Root[]
  (Aux[f='sg', s='ing'] is)
  (NP[f='sg'] (Det[] the) (N[f='sg'] dog))
  (VP[f='ing', -gap] (V[f='ing', s=0, -t] barking)))
```

**c.** `is the dog chasing`

```
Number of trees: 0
```

**d.** `who is the dog chasing`

```
(Root[]
  (Wh[] who)
  (Aux[f='sg', s='ing'] is)
  (NP[f='sg'] (Det[] the) (N[f='sg'] dog))
  (VP[f='ing', +gap] (V[f='ing', s=0, +t] chasing)))
```

**e.** `who the dog is chasing`

```
Number of trees: 0
```

**3.** How do we modify the grammar to handle the following?

**a.** the dog that the cat chases barks

**b.** who do you think the dog chases

**c.** who do you want to chase

**d.** who did you give the present to

**e.** the dog that I gave the present to

**4.** "to whom did you give the present"

**a.** Alternative to ±`gap`:

```
Root -> WhNP       Aux[f=?f, s=?v] NP[f=?f] VP[f=?v, g=NP]
```

**b.** Recall ("give the present to the dog"):

```
VP[f=?f] -> V[f=?f, +t, s=?p] NP PP[f=?p]
```

**c.** Need to keep track of the preposition that was moved ("to whom ... give the present")

```
VP[f=?f, g=?p] -> V[f=?f, +t, s=?p] NP
Root -> WhPP[f=?p] Aux[f=?f, s=?v] NP[f=?f] VP[f=?v, g=?p]
```

**d.** "give the present to"

```
VP[f=?f, g=?g] -> V[f=?f, +t, s=?p] NP PP[f=?p, g=?g]
PP[f=?p, g=NP] -> P[f=?p]
```

**5.** Grammar 2

```
% start Root

###  Grammar
Root -> S[g=0]
Root -> Aux[f=?f, s=?v] NP[f=?f] VP[f=?v, g=0]
Root -> WhNP Aux[f=?f, s=?v] NP[f=?f] VP[f=?v, g=NP]
Root -> WhPP[f=?g] Aux[f=?f, s=?v] NP[f=?f] VP[f=?v, g=?g]
S[g=?g] -> NP[f=?f] VP[f=?f, g=?g]
VP[f=?f, g=?g] -> Aux[f=?f, s=?v] VP[f=?v, g=?g]

VP[f=?f, g=0] -> V[f=?f, -t, s=0]

VP[f=?f, g=0] -> V[f=?f, +t, s=0] NP
VP[f=?f, g=NP] -> V[f=?f, +t, s=0]

VP[f=?f, g=?g] -> V[f=?f, -t, s=?p] PP[f=?p, g=?g]
VP[f=?f, g=?p] -> V[f=?f, -t, s=?p]

VP[f=?f, g=?g] -> V[f=?f, +t, s=?p] NP PP[f=?p, g=?g]
VP[f=?f, g=NP] -> V[f=?f, +t, s=?p]    PP[f=?p, g=0]
VP[f=?f, g=?p] -> V[f=?f, +t, s=?p] NP

NP[f=?f] -> Det[f=?f] N[f=?f]
PP[f=?p, g=0] -> P[f=?p] NP
PP[f=?p, g=NP] -> P[f=?p]
WhPP[f=?p] -> P[f=?p] WhNP

###  Lexicon
Det[f=sg] -> 'a'
Det -> 'the'
N[f=sg] -> 'cat' | 'dog' | 'present'
N[f=pl] -> 'cats' | 'dogs'
Aux[f=sg, s=ing] -> 'is' | 'was'
Aux[f=pl, s=pl] -> 'did' | 'do'
V[f=sg, -t, s=0] -> 'barks'
V[f=pl, -t, s=0] -> 'bark'
V[f=ing, -t, s=0] -> 'barking'
V[f=sg, +t, s=0] -> 'chases'
V[f=ing, +t, s=0] -> 'chasing'
V[f=pl, +t, s=to] -> 'give'
NP -> 'you'
WhNP -> 'who' | 'whom' | 'what'
P[f=to] -> 'to'
P[f=about] -> 'about'
```

**6.** Examples

- **a.** you give the present to the dog
- **b.** did you give the present to the dog
- **c.** (*)did you give the present
- **d.** what did you give to the dog
- **e.** *what did you give the present to the dog
- **f.** to whom did you give the present
- **g.** *to whom did you give to the dog
- **h.** *to whom did you give the present to the dog
- **i.** *about whom did you give the present
- **j.** who did you give the present to
- **k.** *you give the present to

**7.** Slash

- **a.** Kind of gap: `np`, `to`, `about`, ...
- **b.** Special syntax

```
Root -> WhNP V[f=?f, s=?v] NP[f=?f] VP[f=?v]/np
VP[f=?f]/?g -> Aux[f=?f, s=?v] VP[f=?v]/?g
VP[f=?f] -> V[f=?f, +t, s=0] NP
VP[f=?f]/np -> V[f=?f, +t, s=0]
```

- **c.** Alternatively:

```
VP[f=?f]/?g -> V[f=?f, +t, s=0] NP/?g
NP/np ->
```

- **d.** "give the present to"

```
VP[f=?f]/?g -> V[f=?f, +t, s=?p] NP PP[f=?p]/?g
PP[f=?p]/?g -> P[f=?p] NP/?g
```

- **e.** "to whom ... give the present"

```
Root -> WhPP[f=?g] V[f=?f, s=?v] NP[f=?f] VP[f=?v]/?g
VP[f=?f]/?g -> V[f=?f, +t, s=?p] NP PP/?g
PP[f=?p]/?p ->
```

**8.** Grammar 3

```
% start Root

###  Grammar
Root -> S
Root -> Aux[f=?f, s=?v] NP[f=?f] VP[f=?v]
Root -> WhNP Aux[f=?f, s=?v] NP[f=?f] VP[f=?v]/np
Root -> WhPP[f=?g] Aux[f=?f, s=?v] NP[f=?f] VP[f=?v]/?g

S    -> NP[f=?f] VP[f=?f]
S/?g -> NP[f=?f] VP[f=?f]/?g

VP[f=?f]    -> Aux[f=?f, s=?v] VP[f=?v]
VP[f=?f]/?g -> Aux[f=?f, s=?v] VP[f=?v]/?g

VP[f=?f] -> V[f=?f, -t, s=0]

VP[f=?f]    -> V[f=?f, +t, s=0] NP
VP[f=?f]/?g -> V[f=?f, +t, s=0] NP/?g

VP[f=?f]    -> V[f=?f, -t, s=?p] PP[f=?p]
VP[f=?f]/?g -> V[f=?f, -t, s=?p] PP[f=?p]/?g

VP[f=?f]    -> V[f=?f, +t, s=?p] NP PP[f=?p]
VP[f=?f]/?g -> V[f=?f, +t, s=?p] NP PP[f=?p]/?g
VP[f=?f]/?g -> V[f=?f, +t, s=?p] NP/?g PP[f=?p]

NP/np ->
NP[f=?f] -> Det[f=?f] N[f=?f]
PP[f=?p] -> P[f=?p] NP
PP[f=?p]/?g -> P[f=?p] NP/?g
PP[f=?p]/?p ->
WhPP[f=?p] -> P[f=?p] WhNP

###  Lexicon
Det[f=sg] -> 'a'
Det -> 'the'
N[f=sg] -> 'cat' | 'dog' | 'present'
N[f=pl] -> 'cats' | 'dogs'
Aux[f=sg, s=ing] -> 'is' | 'was'
Aux[f=pl, s=pl] -> 'did' | 'do'
V[f=sg, -t, s=0] -> 'barks'
V[f=pl, -t, s=0] -> 'bark'
V[f=ing, -t, s=0] -> 'barking'
V[f=sg, +t, s=0] -> 'chases'
```

```
    V[f=ing, +t, s=0] -> 'chasing'
    V[f=pl, +t, s=to] -> 'give'
    NP -> 'you'
    WhNP -> 'who' | 'whom' | 'what'
    P[f=to] -> 'to'
    P[f=about] -> 'about'
```